```
C*************************************************************
C
C     CB(30,10)   = Concentration in the bubble phase,
C                   moles/cu.cm.
C                   (Stage number, Component number)
C     CE(30,10)   = Concentration in the emulsion phase,
C                   moles/cu.cm.
C                   (Stage number, Component number)
C     CC(30,10)   = Concentration in the cloud phase,
C                   moles/cu.cm.
C                   (Stage number, Component number)
C     NSTAGE      = Number of the last stage
C     HSTAGE      = Height of the last stage,cm
C     IC          = number of components
C     HT(30)      = Height of each stage, cm
C                   (Stage number)
C     UESN        = Last emulsion phase superficial gas
C                   velocity, cm/s
C     UCSN        = Last cloud phase superficial gas
C                   velocity, cm/s
C     UBSN        = Last bubble phase superficial gas
C                   velocity, cm/s
C
C*************************************************************
C
      SUBROUTINE CONPRT(CB,CE,CC,NSTAGE,HSTAGE,IC,HT,UESN,
     +                  UCSN,UBSN)
C
      REAL CB(30,10),CE(30,10),CC(30,10),HT(30),CAVG(10)
C
C Write out the bubble phase profiles
C
      WRITE(6,100)
  100 FORMAT(1X,'BUBBLE PHASE')
      WRITE(6,110) (I,I=1,IC)
  110 FORMAT(1X,' STAGE  HEIGHT  ',10('C',I3,')',6X))
      DO 10 I=1,NSTAGE
         WRITE(6,120) I,HT(I),(CB(I,J),J=1,IC)
  120 FORMAT(1X,I4,2X,F7.2,2X,20(E9.4,2X))
   10 CONTINUE
C
C Write out emulsion phase profiles
C
      WRITE(6,130)
  130 FORMAT(1X,///,2X,'EMULSION PHASE')
      WRITE(6,110)(I,I=1,IC)
      DO 20 I=1,NSTAGE
         WRITE(6,120) I,HT(I),(CE(I,J),J=1,IC)
   20 CONTINUE
C
C Write out cloud phase profiles
C
```

```
      WRITE(6,140)
  140 FORMAT(1X,///,2X,'CLOUD PHASE')
      WRITE(6,110)(I,I=1,IC)
      DO 20 I=1,NSTAGE
          WRITE(6,120)  I,HT(I),(CC(I,J),J=1,IC)
   20 CONTINUE
C
C Calculate outlet gas composition
C
      DO 40 I=1,IC
          CAVG(I)=(CB(NSTAGE,I)*UBSN+CC(NSTAGE,I)*UCSN+
     +             CE(NSTAGE,I)*UESN)/(UBSN+UCSN+UESN)
   40 CONTINUE
C
C Write out this value
C
      WRITE(6,150)
  150 FORMAT(///,2X,'GAS COMPOSITION AT OUTLET')
      WRITE(6,110)(I,I=1,IC)
      WRITE(6,160)  (CAVG(I),I=1,IC)
  160 FORMAT(16X,20(F9.4,2X))
      RETURN
      END
C*************************************************************
C
C Subroutine ENERGY
C
C This subroutine calculates the heat transfer properties
c of the reactor
C
C*************************************************************
C
C     DP   = Particle diameter, cm
C     RKG  = Thermal conductivity of gas, W/m*K
C     CPS  = Heat capacity of solids, kJ/kg*K
C     DR   = Reactor diameter, cm
C     UO   = Initial superficial gas velocity, cm/s
C     VIS  = Viscosity of gas, kg/cm*s
C     DENS = Density of solids, kg/cu.cm.
C     CPG  = Heat capacity of gas, kJ/kg*K
C     DENG = Gas density, kg/cu.cm.
C     H    = Expanded bed height, cm
```

```
C*************************************************************
C
C Program SLURRY
C
C      This program calculates the concentration and
C temperature profiles in a slurry reactor using a
C modified finite difference technique.
C
C*************************************************************
C
C  Declared Variables
C
C    X(3)   = Concentration of hydrogen in the liquid
C             phase, dimensionless (C/Csat,c)
C             (number of stage :  1 = next stage
C                                 2 = current stage
C                                 3 = last stage )
C     Y(3)  = Concentration of hydrogen in the gas phase,
C             dimensionless (C/Co), (same index as X)
C     T(3)  = Temperature of liquid, dimensionless,
C             (T/Tw), (same index as X)
C
C*************************************************************
      REAL X(3),Y(3),T(3)
C*************************************************************
C
C  Common Variables
C
C     DCOM: User supplied data describing the system
C
C        ALPHA    = Volume contraction factor for gas flow
C                   dimensionless
C        U        = CO/H2 usage ratio, dimensionless
C        RI       = Inlet CO/H2 ratio, dimensionless
C        RKFO     = Preexpotential factor, (s*percent
C                   catalyst in slurry)**-1
C        DR       = Reactor diameter, cm
C        H        = Reactor height, cm
C        DP       = Particle diameter, cm
C        TW       = Wall temperature, K
C        UGO      = Initial gas flow rate, cm/s
C        WCAT     = Weight percent catalyst in slurry
C        DENCAT   = Density of the catalyst, g/cu.cm.
C        CPCAT    = Heat capacity of catalyst, cal/g-C
C        CCATM    = Mean catalyst concentration, percent
C        CPL      = Heat capacity of the slurry liquid,
C                   cal/g-C
C        CONL     = Thermal conductivity of the slurry
C                   liquid, J/cm-s-C
```

```
C        P        = Pressure, atm
C        YO       = Initial concentration of hydrogen in
C                   gas stream, mole fraction
C        EA       = Activation energy, kJ/mole
C
C
C     PCOM: Calculated parameters that are independant of
C           bed height.
C
C        ALTWO    = Modified contraction factor
C        UG       = Gas velocity, (Ug/Ugo) dimensionless
C        EG       = Gas holdup, dimensionless
C        DG       = Diffusivity of hydrogen in the gas
C                   sq.cm./s
C        BOG      = Bordenstein number for gas phase, (See
C                   table __ in Barton, Design of Staged..
C                    Thesis), dimensionless
C        BOL      = Bordenstein number for liquid phase,
C                   dimensionless
C        BOC      = Catalyst settling Bordenstien number,
C                   dimensionless
C        A        = Gas-liquid interfacial area, 1/cm.
C        RKF      = Rate constant including stoichiometery
C                   same units as RKFO
C        DA       = Damkohler number, (See Barton 1986)
C        DL       = Diffusivity of hydrogen in the liquid,
C                   sq.cm./s
C        E        = Specific energy dissipation rate,
C                   sq.cm./cu.sec.
C        GAMMA    = Arrhenius number, dimensionless
C        DHR      = Reaction enthalpy, kJ/mole
C        AH       = specific heat transfer area, 1/cm
C
C   HCOM: Calculated parameters that depend on height
C
C        HE       = Henry coefficient, cu.cm. kPa*ml/mol
C        RKL      = Liquid side mass transfer coefficient,
C                   cm/s
C        RKS      = liquid-solid mass transfer coefficient
C                   cm/s
C        STG      = Stanton number in gas phase
C                   dimensionless
C        STL      = Stanton number in liquid phase,
C                   dimensionless
C        STH      = Stanton number for heat transfer
C        CCAT     = Catalyst concentration, percent
C        RNS      = Liquid-solid mass transfer effective-
C                   ness factor, dimensionless
```

```
C         PE       = Peclet number
C         HT       = Heat transfer coefficient to wall
C         BE       = Dimensionless parameter, see Barton
C                    1986
C
C*************************************************************
      COMMON/DCOM/ALPHA,U,RI,RKFO,DR,H,DP,TW,UGO,WCAT,
     +            DENCAT,CPCAT,CCATM,CPL,CONL,P,YO,EA
C
      COMMON/PCOM/ALTWO,UG,EG,DG,BOG,BOL,BOC,A,RKF,DA,
     +            DL,E,GAMMA,DHR,AH
C
      COMMON/HCOM/HE,RKL,RKS,STG,STL,STH,CCAT,RNS,PE,HT,
     +            BE
C
C Obtain initial data for the current reactor
C (See: Subroutine GETDAT)
C
      CALL GETDAT(NSTEP,X,Y,T,YOUT,IP)
C
C Initialize the current height and set the height step
C size
C
      CH=0.0
      DZ=1.0/NSTEP
C
C Calculate the parameters that realte to the current bed
C from the initial data.  (See: Subroutine PARAM)
C
      CALL PARAM(YOUT)
C
C Step through the entire bed.
C
      DO 30 I=1,NSTEP
C
C Update the current height
C
         CH=CH+DZ
C
C Calculate the height dependant parameters. (See:
C Subroutine HPAR)
C
         CALL HPAR(T(2),CH,Y(2),I)
C
C Calculate liquid concentration ( Subroutine XCALC),
C gas concentration (Subroutine YCALC), and temperature
C profile (Subroutine TCALC)
C
```

```
          CALL YCALC(STG,UGO,UG,ALTWO,YO,X,Y,T,DZ)
          CALL XCALC(BOL,STL,DA,RNS,CCAT,GAMMA,YO,X,Y,T,
         +           DZ)
          CALL TCALC(PE,STH,BE,DA,RNS,CCAT,GAMMA,X,T,DZ)
C
C Print out current concentrations
C
          CALL REPORT(X,Y,T,I,IP,NSTEP)
C
C Adjust profile arrays for next step
C
          X(3)=X(2)
          X(2)=X(1)
          Y(3)=Y(2)
          Y(2)=Y(1)
          T(3)=T(2)
          T(2)=T(1)
C
   30 CONTINUE
      STOP
      END
C*************************************************************
C
C  Subroutine GETDAT
C
C     This subroutine obtains data on the properties of
C the bed.
C
C-------------------------------------------------------------
C
C     NSTEP  = The number of stages the reactor is to be
C              divided into
C     X(3)   = Liquid concentrations
C     Y(3)   = Gas concentrations
C     T(3)   = Temperatures
C     YOUT   = Guesses outlet gas concentration
C     IP     = Indicator parameter  IP = 1 :long report
C                                   IP = 0 :short report
C
C  DCOM is apssed to this routine.  See the main program
C  for a description of the parameters.
C
C*************************************************************
      SUBROUTINE GETDAT(NSTEP,X,Y,T,YOUT,IP)
C
      COMMON/DCOM/ALPHA,U,RI,RKFO,DR,H,DP,TW,UGO,WCAT,
     +            DENCAT,CPCAT,CCATM,CPL,CONL,P,YO,EA
C
C Declare arrays
```

```
C
      REAL X(3),Y(3),T(3)
C
C Open the input data file.  This statement is highly
C dependant on the version of FORTRAN used.  It is
C written for IBM Mircosoft FORTRAN 7
C
      OPEN(7,FILE='NSLUR.DAT',STATUS='OLD',
     +     ACCESS='DIRECT',FORM='FORMATTED',RECL=80)
C
C Read in the parameters free format
C
      READ(7,*) ALPHA,U,RI,RKFO,EA
      READ(7,*) DR,H,DP
      READ(7,*) TW,P,YO,UGO
      READ(7,*) WCAT,DENCAT,CPCAT,CCATM
      READ(7,*) CPL,CONL
      READ(7,*) NSTEP
      READ(7,*) (X(I),I=1,3)
      READ(7,*) (Y(I),I=1,3)
      READ(7,*) (T(I),I=1,3)
      READ(7,*) YOUT
      READ(7,*) IP
      RETURN
      END
C*************************************************************
C
C Subroutine Param
C
C  This subroutine calculate the parameters required
C that are not dependant on height
C
C-------------------------------------------------------------
C
C      YOUT  = Guessed outlet concentration, dimension-
C              less.
C  Commons DCOM and PCOM are passed to the subroutine.
C see the main routine for a description of the para-
C meters.
C
C*************************************************************
      SUBROUTINE PARAM(YOUT)
C
      COMMON/DCOM/ALPHA,U,RI,RKFO,DR,H,DP,TW,UGO,WCAT,
     +            DENCAT,CPCAT,CCATM,CPL,CONL,P,YO,EA
C
      COMMON/PCOM/ALTWO,UG,EG,DG,BOG,BOL,BOC,A,RKF,DA,
     +            DL,E,GAMMA,DHR,AH
C
```

```
C Set the constants to be used in the subroutine
C    G = gravitational acceleration, cm./s*s
C    R = gas constant, J/atm*K
C
      G=980.669
      R=0.008314
C
C Calculate the heat of reaction based on stoichiometric
C considerations
C
      DHR=-(165.0+40*(U-0.5)/1.5)
C
C Calculate modified contraction factor .  Accounts for
C initial reactant make up and catalyst usage ratio.
C
     ALTWO=ALPHA*(1+U)/(1+RI)
C
C Calculate velocity based parameters based on bed
C average gas velocity.  Average velocity based on
C guessed outlet composition
C
      YAVG=(1+YOUT)/2.0
      UGA=(1+ALTWO)/(1+ALTWO*YAVG)
C
C Gas hold up.  Deckwer et.al.,Ind.Eng.Chem. Process Des.
C Dev.,1980,19,699.
C
      EG=0.053*POW(UGA,1.1)
C
C Gas phase dispersion coefficient. Mangartz,1980 in
C Deckwer, Ind.Eng.Chem.Proc.Des.Dev.,1982,21,p.241
C
       DG=5.0E-4*(UGA/EG)*(UGA/EG)*(UGA/EG)*DR**1.5
C
C Calculate Bodenstein numbers based on their definitions
C
      BOG=UGO*H/(DG*EG)
      BOL=2.83*POW(UGA*UGA/(G*DR),0.34)
      IF(ABS(BOL).LT.1E-6) BOL=1E-6*BOL/ABS(BOL)
C
C Interfacial area. Deckwer et.al.,Ind.Eng.Chem.Proc.Des.
C Dev.,1980,19,699.
C
      A=4.5*POW(UGA,1.1)
C
C Rate constant taking into account stoichometery
C
      RKF=RKFO/(1+U)
C
C Damkohler number by definition
```

```
C
      DA=RKF*(1-EG)*H/UGO
C
C Liquid phase dispersion coefficient, Shah, Scaleup
C in Chemical Process Industries, ed. Kabel,R.,Wiley,1982
C
      DL=3.676*POW(UGA,0.32)*POW(DR,1.34)
C
C Bordenstein number for liquid
C
      BOL=UGO*H/(DL*(1-EG))
C
C Energy dissipation rate, Sanger, Chem.Eng.J.,1981,22,
C p. 179
C
      IF(UGA.LT.6) THEN
         E=UGA*G
      ELSE
         E=5886.0
      ENDIF
C
C Froude number for calculation of solids settling rate
C
      FR=UGA/SQRT(G*DR)
C
C Specific heat transfer area and arrhenius number
C
      AH=(2*3.14*DR*H)/(3.14*DR*DR*H)
      GAMMA= EA/(R*(TW+273))
      UG=1.0
      RETURN
      END
C***********************************************************
C
C Subroutine HPAR
C
C  Subroutine to calculate the parameters that vary with
C height.
C
C-----------------------------------------------------------
C
C          T = Current temperature
C          Z = Current height
C          Y = Current gas phase concentration
C          I = Number of the current stage
C
C  The commons DCOM, PCOM, and HCOM are passed to this
C subroutine.  See the main routine for a description of
C the parameters.
```

```
C
C*************************************************************
      SUBROUTINE HPAR(T,Z,Y,I)
C
      COMMON/DCOM/ALPHA,U,RI,RKFO,DR,H,DP,TW,UGO,WCAT,
     +            DENCAT,CPCAT,CCATM,CPL,CONL,P,YO,EA
C
      COMMON/PCOM/ALTWO,UG,EG,DG,BOG,BOL,BOC,A,RKF,DA,
     +            DL,E,GAMMA,DHR,AH
C
      COMMON/HCOM/HE,RKL,RKS,STG,STL,STH,CCAT,RNS,PE,HT,
     +            BE
C
C Set the constants need for the routine
C       G =
C    G = gravitational acceleration, cm./s*s
C    R = gas constant, J/atm-K
C
      R=0.008312
      G=980.669
C
C Calcualte the Henry coefficeint, Peter, Z. Phys. Chem.,
C 1955,5,p.114
C
      HE=2.297E7*EXP(-1.2326+583/(T*TW+273))
C
C Calculate the liquid density, (Fischer-Tropsch product)
C Satterfield, Chem. Eng. Sci.,1980,35, p.195
C
      DENL=0.758-0.555E-3*((T*TW)-100)
C
C Liquid viscosity. Deckwer et.al.,Ind.Eng.Chem.Proc.Des.
C Dev.,1980,19,699.
C
      VISL=0.052*EXP(-6.905+3266/(T*TW+273))
C
C Calculate average physical properties of slurry
C
      VCAT=(DENL*WCAT)/(DENCAT-WCAT*(DENCAT-DENL))*WCAT
      DENAVG=VCAT*DENCAT+(1-VCAT)*DENL
      VISAVG=VISL*(1+4.5*VCAT)
C
C Liquid-gas mass transfer coeffiecient, Satterfield,
C Chem. Eng. Sci.,1980,35, p.195
C
      RKL=0.1165*POW(DENAVG/VISAVG*EXP(-4570/
     +    (T*TW+273)),0.333)
C
C Stanton number fromtheir definitions
C
      STG=(RKL*A*H*R*(TW+273))/(UGO*HE)
      STL=RKL*A*H/UGO
```

```
C
C Calculate the effectiveness factor, solid liquid mass
C transfer coefficient based on Sanger, (see above)
C
      AS=6.0*WCAT*(1-EG)*DENAVG/(DP*DENCAT)
      RKS=E*POW(DP,4)*POW(DENL,3)/POW(VISL,3)
      RKS=2+RKS*POW((VISL/(DENL*DL)),0.333)*0.545
      RKS=(DL*RKS)/DP
      RNSO=RKF*CCATM*EL/(RKS*AS)
      RNS=1/(1+RNSO*EXP(-GAMMA/T))
C
C Calculate catalyst concnetration, Kato, J.Chem.Eng.Jpn.
C 1972,5,p.112
C
      CCAT=CCATM*BOL/(1-EXP(-BOL))*EXP(-BOL*Z)
C
C Calcualte average quantities needed for Deckwer, 1980,
C correlation for heat transfer coefficient.
C
      CPAVG=WCAT*CPCAT+(1-WCAT)*CPL
      DISAX=DL*DENAVG*CPAVG
      DIV2=(1.0-EG)*DISAX
      PE=UGO*DENAVG*CPAVG*H/((1-EG)*DISAX)
      CONAVG=CONL*(2*CONL+CONCAT-2*VCAT*(CONL-CONCAT))
      CONAVG=CONAVG/(2*CONL+CONCAT+VCAT*(CONL-CONCAT))
      HT=(UG*UG*UG)*DENAVG/(G*VISAVG)*(VISAVG*CPAVG/
     +   CONAVG)*(VISAVG*CPAVG/CONAVG)
      HT=POW(HT*(VISAVG*CPAVG/CONAVG),-0.25)
      HT=0.1*DENAVG*CPAVG*CPAVG*UG*HT
C
C Stanton number
C
      STH=HT*AH*H/(UGO*DENAVG*CPAVG)
C
C  Dimensionless parameter
C
      BE=DHR*P*YO/(DENAVG*CPAVG*HE*(TW+273))*153
C
C Gas velocity,
C
      IF(I.EQ.1) THEN
         UG=1.0
      ELSE
         XH=(YO-UG*Y)/(YO)
         UG=1+ALPHA*((1+U)/(1+RI)*XH)
      ENDIF
      RETURN
      END
```

```
C*************************************************************
C
C Subroutine YCALC
C
C This subroutine calculates the gas phase concentration
C based on material balance
C
C*************************************************************
      SUBROUTINE YCALC(STG,UGO,UG,ALTWO,YO,X,Y,T,DZ)
      REAL X(3),Y(3),T(3)
      T1=DZ*STG*T(2)*(Y(2)-X(2)*YO)
      Y(1)=Y(2)-T1
      RETURN
      END
C*************************************************************
C
C Subroutine XCALC
C
C This subroutine calculates the liquid phase composition
C based on a material balance on a small but finite area
C
C*************************************************************
      SUBROUTINE XCALC(BOL,STL,DA,RNS,CCAT,GAMMA,YO,X,Y,
     +                 T,DZ)
      REAL X(3),Y(3),T(3)
      T1=1.0/(DZ*BOL)*(X(3)-X(2))
      T2=DZ*STL*(Y(2)/YO-X(2))
      T3=DZ*DA*RNS*CCAT*EXP(-GAMMA/T(2))*X(2)
      X(1)=X(2)-DZ*BOL*(T1+T2-T3)
      RETURN
      END
C*************************************************************
C
C Subroutine TCALC
C
C  This subroutine calculates the temperature based on an
C energy balance
C
C*************************************************************
      SUBROUTINE TCALC(PE,STH,BE,DA,RNS,CCAT,GAMMA,X,T,
     +                 DZ)
      REAL T(3),X(3)
      V1=1.0/(DZ*PE)*(T(3)-T(2))
      V2=DZ*STH*(T(2)-1)
      V3=BE*DA*RNS*CCAT*EXP(-GAMMA/T(2))*X(2)
      T(1)=T(2)+PE*DZ*(V1-V2+V3)
      RETURN
      END
```

```
C*************************************************************
C
C Subroutine REPORT
C
C This subroutine lists the current concentrations.
C
C*************************************************************
      SUBROUTINE REPORT(X,Y,T,I,IP,NSTEP)
      REAL T(3),X(3),Y(3)
C
C if a short report was indicated list only the final
C three concnetrations
C
      IF((IP.EQ.1).AND.(I.LE.NSTEP-3)) GOTO 20
      IF(I.NE.1) GOTO 10
      WRITE(6,100)
  100 FORMAT(' I          T          X          Y')
   10 CONTINUE
      WRITE(6,110) I,T(1),X(1),Y(1)
  110 FORMAT(1X,I3,3F10.4)
   20 CONTINUE
      RETURN
      END
C*************************************************************
C
C Function POW
C
C This function raises any number to any power.  It was
C written to the inability of the intrinsic function to
C handle negative numbers.
C
C*************************************************************
C
C     BASE  = The number to be raised to a power
C     POWER = The number to be used as the power
C
C*************************************************************
C
      FUNCTION POW(BASE,POWER)
C
C If the base is a positive number set SIGN=1.0 so that
C the answer
C will be positive.
C
      IF (BASE.EQ.0.0) THEN
         POW=0.0
         RETURN
      ENDIF
      IF(BASE.GT.0.0) THEN
         A=POWER*ALOG(BASE)
```

```
         SIGN=1.0
      ELSE
C
C If the base is negative and the power is even then the
C answer will be positive.  If the base is negative and
C the power is not even the answer is to be negative.
C
          REM=(POWER/2.0)-INT(POWER/2.0)
          A=POWER*ALOG(ABS(BASE))
        IF(REM.EQ.0.0) THEN
           SIGN=1.0
        ELSE
           SIGN=-1.0
        ENDIF
      ENDIF
      POW=SIGN*EXP(A)
      RETURN
      END
```

```
C*************************************************************
C
C Program COMB
C
C    This program calculates the heat required by the
C combustor and the temperature of the combustor.
C
C*************************************************************
C
C  Common block
C
C          QP    = Heat load of the pyrolyzer, cal
C          AE    = External heat transfer area, sq.cm.
C          TE    = External temperature, K
C          AI    = Heat transfer area between combustor
C                  and pyrolyzer, sq.cm.
C          TP    = Pyrolyzer temperature. K
C          CS    = Heat capacity of the solid material,
C                  cal/g K
C          DENS  = Density of the solids, g/cu.cm.
C          CG    = Heat capacity of the gas, cal/g K
C          DENG  = Density of the gas, g/cu.cm.
C          VIS   = Viscosity of the gas,
C          RKG   = Thermal conductivity of the gas,
C          DP    = Partical diameter, cm
C          EMF   = Void fraction at minimum fluidization.
C          DR    = Reactor diameter, cm.
C
C*************************************************************
      COMMON QP,AE,TE,AI,TP,CS,DENS,CG,DENG,VIS,RKG,DP,
     +       EMF,DR
C
C Open the data file.  This statement is dependant on the
C the FORTRAN version used.  The statement listed is for
C a IBM Microsoft FORTRAN
C
      OPEN(5,FILE='COMB.DAT',STATUS='OLD',ACCESS='DIRECT',
     +     FORM='FORMATTED',RECL=80)
C
C  Obtain the initial parameters.  (See: Subroutine
C PARAM)
C
      CALL PARAM(VGC,VSP,HMFC,HC,HMFP,HP,VS)
C
C Calculate the internal (Subroutine RINTER) and external
C (Subroutine EXTER) heat transfer coefficients
```

```
C
      UE=EXTER(VGC,HC,HMFC)
      UI=RINTER(VGC,VGP,HC,HP,HMFC,HMFP)
C
C Calculate the required combustor temperature to
C provide required heat to pyrolyzer at the required
C temperature.
C
      TC=TP+QP/(UI*AI+CS*DENS*VS)
C
C Calculate the heat losses from the combustor.
C     Q1=heat lost through wall
C     Q2=heat lost in gas
C
      Q1=UE*AE*(TC-TE)
      Q2=CG*DENG*VGC*TC
C
C Sum for total heat requirment
C
      Q=Q1+Q2+QP
C
C Write out result
C
      WRITE(0,100) TC,Q
  100 FORMAT('  T   Q ',2E12.4)
      STOP
      END
C*************************************************************
C
C.Subroutine Param
C
C   This subroutine reads in the needed data from the
C input file.
C
C*************************************************************
C
C                VGC  = Superficial velocity of the gas
C                       in the combustor, cm/s
C                VGP  = Superficial velocity of the gas
C                       in the pyrolyzer, cm/s
C                HMFC = Height of the combustor bed at
C                       minimum fluidization, cm
C                HMFP = Height of the pyrolyzer bed at
C                       minimum fluidization, cm
C                HP   = Height of the pyrolyzer, cm
C                HC   = Height of the combustor, cm
```

```
C                VS    = Velocity of solids circulation
C
C  The common block is passed to the subroutine.  See the
C main routine for a description of the parameters in the
C common.
C
C*************************************************************
      SUBROUTINE PARAM(VGC,VGP,HMFC,HC,HMFP,HP,VS)
C
      COMMON QP,AE,TE,AI,TP,CS,DENS,CG,DENG,VIS,RKG,
     +       DP,EMF,DR
C
      READ(5,*) DC,TE,DR,TP
      READ(5,*) QP
      READ(5,*) CS,DENS,VS
      READ(5,*) CG,DENG,VGC,VGP
      READ(5,*) HMFC,HC,HMFP,HP
      READ(5,*) VIS,RKG,DP,EMF
      AE=DC*3.1416*HC
      AI=DR*3.1416*HP
      RETURN
      END
C*************************************************************
C
C Function EXTER
C
C   This function calculates the external heat transfer
C coefficient
C
C*************************************************************
C
C        VG  =.Velocity of gas in the combustor, cm/s
C        H   = Total bed height, cm
C        HMF = Height of bed at minimum fluidization, cm
C
C   The common block is passed to the function.  See the
C main routine for a description of the parameters.
C
C*************************************************************
      FUNCTION EXTER(VG,H,HMF)
C
      COMMON QP,AE,TE,AI,TP,CS,DENS,CG,DENG,VIS,RKG,DP,
     +       EMF,DR
C
C HEW1= Resistance to heat transfer by natural convection
C       off the outside of the reactor.
```

```
C HEW2= Resistance from the fluidized bed to the wall
C
      HEW1=(0.067544*0.59/DR)*(0.702*0.4462E9*TE*(DR**3))**0.25
      HEW2=FLUID(VG,H,HMF)
      EXTER=1.0/(1.0/HEW1+1.0/HEW2)
      RETURN
      END
C*************************************************************
C
C Function RINTER
C
C     This function calculates the coefficient for the
C inner wall.
C
C*************************************************************
C
C                VGC  = Superficial velocity of the gas
C                       in the combustor, cm/s
C                VGP  = Superficial velocity of the gas
C                       in the pyrolyzer, cm/s
C                HMFC = Height of the combustor bed at
C                       minimum fluidization, cm
C                HMFP = Height of the pyrolyzer bed at
C                       minimum fluidization, cm
C                HP   = Height of the pyrolyzer, cm
C                HC   = Height of the combustor, cm
C
C  The common block is passed to the subroutine.  See the
C main routine for a description of the parameters in the
C common.
C
C*************************************************************
      FUNCTION RINTER(VGC,VGP,HC,HP,HMFC,HMFP)
C
      COMMON QP,AE,TE,AI,TP,CS,DENS,CG,DENG,VIS,RKG,DP,
     +       EMF,DR
C
C HEW1 = Resistance on the combustor side of the wall
C HEW2 = Resistance on the pyrolyzer side of the wall
C
      HEW1=FLUID(VGC,HC,HMFC)
      HEW2=FLUID(VGP,HP,HMFP)
      RINTER=1.0/(1.0/HEW1+1.0/HEW2)
      RETURN
      END
```

```
C*************************************************************
C
C Function FLUID
C
C     This function calculates the heat transfer
C resistance of a wall in contact with a fluidized bed.
C
C*************************************************************
C
C       VG  = Velocity of the gas, cm/s
C       H   = Total bed height, cm
C       HMF = Bed height at minimum fluidization, cm
C
C  The common block is passed to the subroutine.  See the
C main routine for a description of the parameters in the
C common.
C
C*************************************************************
      FUNCTION FLUID(VG,H,HMF)
C
      COMMON QP,AE,TE,AI,TP,CS,DENS,CG,DENG,VIS,RKG,DP,
     +       EMF,DR
C
C Look up the parameter Si used in the correlation
C
      SI=SITBL(DP,DENG,VG,VIS)
C
      HF=1.0-((1.0-EMF)*HMF/H)
      AT=(1.0+7.5*EXP(-0.44*(H/DR)*(CG/CS)))
      BT=(DP/RKG)/((1-HF)*CS*DENS/(CG*DENG))
      FLUID=(SI*AT)/BT
      RETURN
      END
C*************************************************************
C
C Function SITBL
C
C    This function determines the parameter Si
C
C*************************************************************
C
C         DP   = Partical diameter, cm
C         DENG = Density of the gas, g/cu.cm.
C         VG   = Gas velocity, cm/s
C         VIS  = Viscosity of the gas,
C
```

```
C*************************************************************
      FUNCTION SITBL(DP,DENG,VG,VIS)
C
      RE=DP*DENG*VG/VIS
      RELOG=ALOG(RE)
      IF(RELOG.GT.-0.66) GOTO 10
      A=1.08
      B=-2.99
      GOTO 50
   10 IF(RELOG.GT.0.0) GOTO 20
      A=0.91
      B=-2.99
      GOTO 50
   20 IF(RELOG.GT.0.9) GOTO 30
      A=0.64
      B=-3.10
      GOTO 50
   30 IF(RELOG.GT.1.48) GOTO 40
      A=0.25
      B=-2.71
      GOTO 50
   40 A=0.0
      B=-2.3
   50 SILOG=A*RELOG+B
      SITBL=10**SILOG
      RETURN
      END
```